

ECLS-K 1998-99 Food Security Status File Technical Documentation and User Notes

Data File and Documentation Prepared by
Mark Nord
Economic Research Service
United States Department of Agriculture
January 26, 2001

Revision history of this document:

January 26, 2001--original creation of document

February 8, 2001-- revised

Background

The ECLS-K Food Security Status Data File (hereafter Food Security Status File) contains summary food security status information for households of children in the Early Childhood Longitudinal Study, Kindergarten Class of 1998-99 (ECLS-K). The summary food security status variables are based on responses to the 18 food security questions (the food security core module), which were collected in interviews of the children's parents in spring 1999 survey (Spring 2). *The food security interview data were accessed from the ECLS-K Child Data File (NCES 2001-029) to create the Food Security Status File, which matches to the Child Data File by CHILDDID.*

The Food Security Status File contains only the food security summary variables and CHILDDID variable. It is intended to be used in conjunction with the ECLS-K Child, Teacher, and School files, which contain extensive information on children and their schools and families. These data files, along with full documentation and other information on the study are available from the National Center for Education Statistics. For access information, go to the NCES website: <http://nces.ed.gov>.

Technical Description

The Food Security Status File is an ASCII file containing 21,260 records. The length of each record is 16 characters. The data dictionary is included below as Appendix A. SAS code to read the ASCII data file and create a SAS file is included as Appendix B. Frequency tabulations of each of the three food security status variables are included as Appendix C. SAS code to create the food security composite variables directly from the ECLS-K Child Data File is in Appendix D.

Matching to the ECLS-K 1998-99 Child Data File

The Food Security Status File contains one record for each child (record) in the ECLS-K 1998-99 Child Data File (NCES 2001-029). It matches to that data file by CHILDDID. Both the Food Security Status File and the ECLS-K Child data files are already sorted by CHILDDID.

Food Security Status Variables

Food security status variables were calculated based on the 18 core items in the food security module, P2WORRFD through P2NOMONY. Calculations were carried out in accordance with the standard methods described in *Guide to Measuring Household Food Security, Revised 2000*.¹ Assessment of the food security items using Rasch measurement model methods indicated that relative item severities were similar to those in the 1998 Current Population Survey Food Security Supplement, so the use of the standard benchmark household scores was appropriate.

¹ Gary Bickel, Mark Nord, Cristofer Price, William Hamilton, and John Cook. *Guide to Measuring Household Food Security, Revised 2000*. U.S. Department of Agriculture, Food and Nutrition Service, Alexandria, VA, March 2000. Available on the Food and Nutrition Service Website, <http://www.fns.usda.gov/oane/>

Three food security status variables are provided as follows:

P2FSRAW is the food security raw score, a simple count of the number of food security items affirmed by the parent.

P2FSSCAL is the food security scale score. This is a measure of the severity of food insecurity or hunger experienced in the household in the previous 12 months. It is a continuous, interval-level measure based on the Rasch measurement model and is appropriate for linear models, such as correlation, regression, or analysis of variance. It is on the standard (logistic-unit) computational metric described in *Measuring Household Food Security*. Valid values range from 1.4 to 13.0, with higher values indicating more severe food deprivation. Technically, the scale score is undefined for households that affirmed no items. These households are food secure, but the appropriate size of the interval between their score and the score of households that affirmed one item is not known and varies from household to household. The variable is coded -6 for households that affirmed no items (or were screened out and deemed to be food secure) to remind users that these cases require special consideration in analyses.

P2FSSTAT is a categorical measure of food security status that identifies households as food secure, food insecure without hunger, food insecure with hunger (moderate), and food insecure with hunger (severe). This variable is appropriate for comparing prevalence rates of food insecurity and hunger across subpopulations. There were few cases in the most severe category, and for most prevalence reporting purposes, the two categories of food insecure with hunger should be collapsed and reported as a single category. When interpreting food security statistics, users should remember that food security status is a household-level characteristic. In most households classified as food insecure with hunger, the children in the household were not hungry.

Cases With No Valid Responses to Food Security Status Variables

Of the 21,260 records in the ECLS-K 1998-99 Child Data File, 2,298 had no food security data (all data fields were blank). An additional 115 records had some food data but had no valid response to any of the 18 food security scale variables. The food security status of the households of these 2,413 children, therefore, is unknown, and the 3 food security variables for them are coded missing (-9).

Imputation of Missing Responses

No imputation was carried out for missing responses. Excluding the cases that had no valid responses, and considering as valid those responses to questions that were skipped because of screening, only 48 cases (0.25 percent) had any items missing. Not imputing treats missing responses as if they were negative responses (i.e., effectively imputes them as negatives). Analysis of the 48 cases with missing responses revealed that imputation in accordance with procedures specified in *Measuring Household Food Security* would change raw scores of only five cases and would change the food security status category of only one case.

Appendix A

Data Dictionary/Record Layout

Data dictionary for food security variables, ECLS-K 1998-99 Child File (in CPS format)

```
*****
*      CHILD FILE MATCH VARIABLE      *
*****
DATA      SIZE      BEGIN
D CHILDDID      8      1
T Child File case ID
U All cases
V      .ID alphanumeric

*****
*      FOOD SECURITY STATUS VARIABLES      *
*****

D P2FSRAW      2      9
T Food security raw score (household)
T      (number of affirmative responses)
U All cases
V      -9      .Missing, no valid responses
V      0      .No affirmative responses
V      1:18      .Number of affirmative responses

D P2FSSCAL      4      11
T Food security scale score (Rasch score for household)
T      (1 decimal place, decimal point in ascii file)
U All households
V      -9      .Missing, no valid responses
V      -6      .No affirmative responses
V      .scale score undefined
V 1.4:13.0 .Number

D P2FSSTAT      2      15
T Food security status category (household)
U All casess
V      -9      .Missing, no valid responses
V      1      .Food secure
V      2      .Food insecure without hunger
V      3      .Food insecure with hunger (moderate)
V      4      .Food insecure with hunger (severe)
```

Appendix B

SAS Program To Read ECLS-K 1998-99 Food Security Data File

Note: The following SAS code can be accessed in text format from the ECLS-K data page in the ERS Food Security Briefing Room.

```
*read ascii file;
data temp;
infile 'd:\foodsecu\eccls\ecclsfood.dat' lrecl=16; *change to identify path and file in your system;
input
@1 childid $ 8.
@9 p2fsraw 2.
@11 p2fsscal f4.1
@15 p2fsstat 2.;
run;

proc freq data=temp;
tables p2fsraw p2fsscal p2fsstat;
title1 'Run 4: Cases, unweighted';
run;
```

Appendix C

Frequencies of Variables in ECLS-K 1998-99 Food Security Data File

Run 4: Cases, unweighted

08:03 Tuesday, December 5, 2000 277

The FREQ Procedure

p2fsraw	Frequency	Percent	Cumulative Frequency	Cumulative Percent
-9	2413	11.35	2413	11.35
0	15512	72.96	17925	84.31
1	1054	4.96	18979	89.27
2	576	2.71	19555	91.98
3	429	2.02	19984	94.00
4	285	1.34	20269	95.34
5	250	1.18	20519	96.51
6	276	1.30	20795	97.81
7	104	0.49	20899	98.30
8	102	0.48	21001	98.78
9	73	0.34	21074	99.13
10	66	0.31	21140	99.44
11	43	0.20	21183	99.64
12	34	0.16	21217	99.80
13	22	0.10	21239	99.90
14	8	0.04	21247	99.94
15	4	0.02	21251	99.96
16	2	0.01	21253	99.97
17	4	0.02	21257	99.99
18	3	0.01	21260	100.00

p2fsscal	Frequency	Percent	Cumulative Frequency	Cumulative Percent
-9	2413	11.35	2413	11.35
-6	15512	72.96	17925	84.31
1.4	1054	4.96	18979	89.27
2.6	576	2.71	19555	91.98
3.4	429	2.02	19984	94.00
4.1	285	1.34	20269	95.34
4.8	250	1.18	20519	96.51
5.4	276	1.30	20795	97.81
6	104	0.49	20899	98.30
6.6	102	0.48	21001	98.78
7.2	73	0.34	21074	99.13
7.7	66	0.31	21140	99.44
8.3	43	0.20	21183	99.64
8.8	34	0.16	21217	99.80
9.3	22	0.10	21239	99.90
9.8	8	0.04	21247	99.94
10.4	4	0.02	21251	99.96
11.1	2	0.01	21253	99.97
12.2	4	0.02	21257	99.99
13	3	0.01	21260	100.00

p2fsstat	Frequency	Percent	Cumulative Frequency	Cumulative Percent
-9	2413	11.35	2413	11.35
1	17142	80.63	19555	91.98
2	1344	6.32	20899	98.30
3	318	1.50	21217	99.80
4	43	0.20	21260	100.00

Appendix D

SAS Program To Create Food Security Composite Variables Directly from the ECLS-K 1998-99 Child Data File

Note: The following SAS code can be accessed in text format from the ECLS-K data page in the ERS Food Security Briefing Room.

```
*read 18 scale variables and create food security composite variables;
data temp.eclsfood (keep=chldid p2fsraw p2fsscal p2fsstat);
infile 'f:\chld.dat' lrecl=5305;
input #1
  chldid $    1-8
  p2worrfd 4058-4059
  p2fdlast 4060-4061
  p2blmeal 4062-4063
  p2lowcst 4064-4065
  p2noba1 4066-4067
  p2cantaf 4068-4069
  p2evcut2 4070-4071
  p2evcut 4072-4073
  p2eatles 4074-4075
  p2hungry 4076-4077
  p2losewt 4078-4079
  p2noteat 4080-4081
  p2notea2 4082-4083
  p2cutml 4084-4085
  p2chskip 4086-4087
  p2oftcut 4088-4089
  p2chievr 4090-4091
  p2nomony 4092-4093
  #2 #3;

array qfood{18} p2worrfd p2fdlast p2blmeal p2lowcst p2noba1 p2cantaf p2evcut2 p2evcut
  p2eatles p2hungry p2losewt p2noteat p2notea2 p2cutml p2chskip p2oftcut p2chievr p2nomony;
array hhsc{18} hhsc1-hhsc18; *household scores corresponding to raw scores;

*set household scores for each raw score;
*based on Guide to Measuring Household Food Security, 2000;
*standard (logit-unit) computational metric, appendix C-2, page 71;
hhsc1=1.4; hhsc2=2.6; hhsc3=3.4; hhsc4=4.1; hhsc5=4.8; hhsc6=5.4;
hhsc7=6.0; hhsc8=6.6; hhsc9=7.2; hhsc10=7.7; hhsc11=8.3; hhsc12=8.8;
hhsc13=9.3; hhsc14=9.8; hhsc15=10.4; hhsc16=11.1; hhsc17=12.2; hhsc18=13.0;

*identify valid cases and accumulate affirmative responses;
cvalid='no';
p2fsraw=0;
do i=1 to 18; *each item;
  if qfood{i} gt 0 then do; *valid response;
    cvalid='ok';
    if qfood{i} eq 1 then p2fsraw=p2fsraw+1;
    else if i in (1,2,3,4,5,6,8,13,16) and qfood{i} eq 2 then p2fsraw=p2fsraw+1;
  end; *of valid response;
end; *of each item;

*assign scale scores and categorical variable;
if cvalid eq 'no' then do; *no valid response;
  p2fsraw=-9;
  p2fsscal=-9;
  p2fsstat=-9;
```

```
end; *of no valid response;
else do; *1 or more valid response;
  if p2fsraw eq 0 then p2fsscal=-6;
  else p2fsscal=hhsc{p2fsraw};
  if p2fsraw ge 13 then p2fsstat=4;
  else if p2fsraw ge 8 then p2fsstat=3;
  else if p2fsraw ge 3 then p2fsstat=2;
  else p2fsstat=1;
end; *of one or more valid response;
run;
```

End of Document